Math 132
Fall 2006 Exam III

1. Calculate $\displaystyle\int_{-1}^{1} \frac{x}{5+2x+x^2}\,dx$.

a) $\ln(2)-\frac{\pi}{2}$ b) $\frac{\ln(2)}{4}-\frac{\pi}{6}$ c) $\frac{\ln(2)}{2}-\frac{\pi}{8}$ d) $\frac{\ln(2)}{2}-\frac{3\pi}{4}$ e) $\ln(2)-\frac{4\pi}{3}$

f) $\frac{\ln(2)}{2}-\frac{\pi}{4}$ g) $3\ln(2)-\frac{\pi}{3}$ h) $2\ln(2)-\frac{\pi}{4}$ i) $2\ln(2)-\frac{3\pi}{4}$ j) $2\ln(2)-\frac{4\pi}{3}$

Solution: c

```
> with(student):
> J := Int(x/(5+2*x+x^2),x = -1 .. 1);
```

$$J := \int_{-1}^{1} \frac{x}{5+2x+x^2}\,dx$$

```
> A := numer(integrand(J));   B := denom(integrand(J));
```

$$A := x$$

$$B := 5+2x+x^2$$

```
> C := completesquare(B,x);
```

$$C := (x+1)^2+4$$

```
> eqn := x + 1 = 2*tan(t);
```

$$eqn := x+1 = 2\tan(t)$$

```
> J1 := changevar(eqn, J, t);
```

$$J1 := \int_{0}^{\frac{\pi}{4}} \frac{2(-1+2\tan(t))(1+\tan(t)^2)}{3+4\tan(t)+(-1+2\tan(t))^2}\,dt$$

```
> J2 := map(simplify, J1 );
```

$$J2 := \int_0^{\frac{\pi}{4}} -\frac{1}{2} + \tan(t)\, dt$$

```
> antiderivative := int(integrand(J2), t);
```

$$antiderivative := -\frac{t}{2} - \ln(\cos(t))$$

```
> simplify(subs(t=Pi/4, antiderivative) - subs(t=0,
  antiderivative));
```

$$-\frac{\pi}{8} + \frac{1}{2}\ln(2)$$

```
> int(x/(5+2*x+x^2),x = -1 .. 1);
```

$$\frac{1}{2}\ln(2) - \frac{\pi}{8}$$

2. The triangular region in the first quadrant bounded by $y = 2x$, $y = x$, and $x = 1$ is rotated about the line
y = -1. What is the volume of the solid of revolution that results?

a) $\frac{\pi}{4}$ b) $\frac{\pi}{2}$ c) $\frac{3\pi}{4}$ d) π e) $\frac{4\pi}{3}$

f) $\frac{3\pi}{2}$ g) $\frac{5\pi}{3}$ h) 2π i) 3π j) 4π

Solution: h

```
> volume := Pi*int((1+2*x)^2-(1+x)^2, x = 0 .. 1);
```

$$volume := 2\pi$$

3. Calculate the arc length of $y = \frac{x^2}{2} - \frac{\ln(x)}{4}$ for $1 \le x \le 2$.

a) $\frac{1}{2} + \frac{\ln(2)}{3}$ b) $\frac{2}{3} + \frac{\ln(2)}{4}$ c) $\frac{3}{2} + \frac{\ln(2)}{4}$ d) $\frac{3}{4} + \frac{\ln(2)}{2}$ e) $\frac{4}{3} + \frac{\ln(2)}{2}$

f) $\frac{1}{2} - \frac{\ln(2)}{3}$ g) $\frac{2}{3} - \frac{\ln(2)}{4}$ h) $\frac{3}{2} - \frac{\ln(2)}{4}$ i) $\frac{3}{4} - \frac{\ln(2)}{2}$ j) $\frac{4}{3} - \frac{\ln(2)}{2}$

Solution: c

```
> f := x -> x^2/2-ln(x)/4;
```

$$f := x \rightarrow \frac{1}{2}x^2 - \frac{1}{4}\ln(x)$$

```
> toBeIntegrated := simplify(sqrt(1+diff(f(x),x)^2))  assuming x
  > 0;
```

$$toBeIntegrated := \frac{1+4x^2}{4x}$$

```
> int(toBeIntegrated, x=1..2);
```

$$\frac{3}{2} + \frac{1}{4}\ln(2)$$

4. The graph of $y = \frac{x^3}{3}$ for $0 \le x \le 1$ is rotated about the x-axis. What is the surface area of the resulting figure?

a) $\frac{\pi(\sqrt{2}-1)}{3}$ **b)** $\frac{\pi(\sqrt{2}-1)}{6}$ **c)** $\frac{\pi(\sqrt{2}-1)}{9}$ **d)** $\frac{\pi(2\sqrt{2}-1)}{3}$ **e)** $\frac{\pi(2\sqrt{2}-1)}{6}$

f) $\frac{\pi(2\sqrt{2}-1)}{9}$ **g)** $\frac{\pi(3\sqrt{2}-1)}{3}$ **h)** $\frac{\pi(3\sqrt{2}-1)}{6}$ **i)** $\frac{\pi(3\sqrt{2}-1)}{9}$ **j)** $\frac{\pi(4\sqrt{2}-1)}{3}$

Solution: f

```
> f := x ->  x^3/3;
```

$$f := x \rightarrow \frac{1}{3}x^3$$

```
> answer := 2*Pi*int(f(x)*sqrt(1+diff(f(x),x)^2),x=0..1);
```

$$answer := 2\pi\left(\frac{\sqrt{2}}{9} - \frac{1}{18}\right)$$

5. Let R be the region bounded by the graph of $y = \sqrt{x}$, the x-axis, and the vertical line $x = 4$. What is the x-coordinate of the center of mass of R?

a) 3 **b)** $\frac{7}{4}$ **c)** $\frac{11}{4}$ **d)** $\frac{14}{5}$ **e)** $\frac{10}{3}$

f) $\frac{9}{5}$ g) $\frac{9}{4}$ h) $\frac{12}{5}$ i) $\frac{5}{3}$ j) $\frac{8}{3}$

Solution: h

```
> x_bar := int(x*sqrt(x),x=0..4)/int(sqrt(x),x=0..4);
```

$$x\_bar := \frac{12}{5}$$

6. What is the y-coordinate of the center of mass of the region R of the preceding question?

a) 1 b) $\sqrt{2}$ c) $\frac{1}{\sqrt{2}}$ d) $\sqrt{3}$ e) $\frac{2}{\sqrt{3}}$

f) $\frac{2}{5}$ g) $\frac{3}{5}$ h) $\frac{2}{3}$ i) $\frac{3}{4}$ j) $\frac{3}{2}$

Solution: i

```
> int(sqrt(x)^2,x=0..4)/2/int(sqrt(x),x=0..4);
```

$$\frac{3}{4}$$

7. Suppose that $f(x) = \sqrt{x}$. For what c is $f(c)$ equal to the average value of $f(x)$ over the interval $[0, 9]$?

a) $\sqrt{5}$ b) 3 c) 4 d) 5 e) $3\sqrt{3}$

f) $\sqrt{6}$ g) $2\sqrt{5}$ h) $4\sqrt{3}$ i) $2\sqrt{2}$ j) $4\sqrt{2}$

Solution: c

```
> a :=0; b := 9; f := x -> sqrt(x);
```

$$a := 0$$

$$b := 9$$

$$f := \text{sqrt}$$

```
> f_ave := int(f(x),x = a .. b)/(b-a);
```

$$f\_ave := 2$$

```
> solve(f(c)=f_ave,c);
```

$$4$$

8. For what constant c is $f(x) = c\,x\,(2 - x)$ a probability density function for $0 \le x \le 2$.

a) $\frac{1}{4}$ b) $\frac{1}{2}$ c) $\frac{2}{3}$ d) 1 e) 2

f) $\frac{4}{3}$ g) $\frac{1}{3}$ h) $\frac{3}{4}$ i) $\frac{3}{2}$ j) 3

Solution: h

```
> eqn := int(c*x*(2-x), x=0..2) = 1;
```

$$eqn := \frac{4\,c}{3} = 1$$

```
> c = solve(eqn, c);
```

$$c = \frac{3}{4}$$

9. A random variable assumes values in the interval $[0, 1]$ and has probability density function $f(x) = \frac{4}{\pi\,(1 + x^2)}$ for $0 \le x \le 1$. What is the mean of X?

a) $\frac{2}{\pi}$ b) $\frac{\sqrt{2}}{\pi}$ c) $\frac{2\sqrt{2}}{\pi}$ d) $\frac{1 - \ln(2)}{\pi}$ e) $\frac{\ln(2)}{2\,\pi}$

f) $\frac{\ln(2)}{\pi}$ g) $\frac{2\ln(2)}{\pi}$ h) $\frac{3 - \ln(2)}{\pi}$ i) $\frac{2\ln(2) - 1}{\pi}$ j) $\frac{e}{\pi}$

Solution: g

```
> int(4/(1+x^2)/Pi,x = 0 .. 1);
```

$$1$$

```
> eqn := mu = Int(4*x/(1+x^2)/Pi, x = 0 .. 1);
```

$$eqn := \mu = \int_0^1 \frac{4\,x}{(1+x^2)\,\pi}\,dx$$

```
> map(value, eqn);
```

$$\mu = \frac{2\ln(2)}{\pi}$$

10. A random variable X assumes values in the interval $[0, \pi]$ and has probability density function $f(x) = \frac{\sin(x)}{2}$ for $0 \le x \le \pi$. What is the value of $P\left(\frac{\pi}{4} \le X \le \frac{\pi}{3}\right)$?

a) $\frac{\sqrt{2}-1}{2}$ b) $\frac{\sqrt{2}-1}{4}$ c) $\frac{\sqrt{3}-1}{2}$ d) $\frac{\sqrt{3}-1}{4}$ e) $\frac{2-\sqrt{2}}{2}$

f) $\frac{2\sqrt{2}-1}{2}$ g) $\frac{2\sqrt{2}-1}{4}$ h) $\frac{2\sqrt{3}-1}{2}$ i) $\frac{2\sqrt{3}-1}{4}$ j) $\frac{\sqrt{3}-\sqrt{2}}{4}$

Solution: b

```
> int(sin(x)/2,x = Pi/4 .. Pi/3);
```

$$-\frac{1}{4}+\frac{\sqrt{2}}{4}$$

11. A random variable X has pdf given by $f(x) = \frac{3\,x^2}{124}$ for $1 \le x \le 5$.

If m is the median of X, then m^3 equals:

a) 58 b) 59 c) 60 d) 61 e) 62

f) 63 g) 64 h) 65 i) 66 j) 67

Solution: f

```
> eqn := Int( 3*x^2/124, x = 1..m) = 1/2;
```

$$eqn := \int_1^m \frac{3\,x^2}{124}\,dx = \frac{1}{2}$$

```
> eqn2 := map(value, eqn);
```

$$eqn2 := \frac{m^3}{124} - \frac{1}{124} = \frac{1}{2}$$

```
> m^3 = solve(eqn2, m^3);
```

$$m^3 = 63$$

12. A spring is maintained 2m beyond equilibrium by a 60N force. If, starting from equilibrium, 120J of work have been expended stretching a spring, then how many meters beyond equilibrium has it been stretched?

a) $\sqrt{2}$ b) $\sqrt{3}$ c) 2 d) $\sqrt{5}$ e) $\sqrt{6}$
f) $\sqrt{7}$ g) $2\sqrt{2}$ h) 3 i) $\sqrt{10}$ j) $2\sqrt{3}$

Solution: g

```
> HookesLaw := F = k*x;
```

$$HookesLaw := F = k\,x$$

```
> eqn := subs({F=60, x=2}, HookesLaw);
```

$$eqn := 60 = 2\,k$$

```
> eqn2a := k = solve(eqn, k);
  eqn2b := W = 120;
```

$$eqn2a := k = 30$$

$$eqn2b := \mathrm{W} = 120$$

```
> eqn3 := W = int(k*x, x = 0..a);
```

$$eqn3 := \mathrm{W} = \frac{k\,a^2}{2}$$

```
> eqn4 := subs( {eqn2a, eqn2b}, eqn3);
```

$$eqn4 := 120 = 15\,a^2$$

```
> a = solve(eqn4, a);
```

$$a = (-2\sqrt{2}, 2\sqrt{2})$$

13. The chemistry department has a cylindrical tank that has height 4m, base radius 2m, and that is partially filled with a fluid that has been concocted to have a weight density of $\frac{1}{\pi}\frac{N}{m^3}$.

If the initial depth of the fluid is 3m and 1/3 of the initial volume is pumped to the top of the tank, then how many Joules of work have been done?

a) 2 b) 3 c) 4 d) 6 e) 8
f) 9 g) 12 h) 16 i) 18 j) 24

Solution: d

```
> A := int(Pi*(1/Pi)*y*4,y = 1 .. 2);
```

$$A := 6$$

14. Forty feet of a uniform cable hang over the side of a building. The cable weighs 6 lbs/ft. A 48 pound hook is attached to the dangling end of the cable. The cable is pulled up 10 feet before the hook snags on a flag pole. How many foot-pounds of work have been done to that point?

a) 2500 b) 2520 c) 2540 d) 2560 e) 2580
f) 2600 g) 2620 h) 2640 i) 2660 j) 2680

Solution: e

```
> 48*10+ 30*6*10+int(6*y,y = 0 .. 10);
```

$$2580$$

15. Calculate $\int_1^5 \frac{1}{\sqrt{5-x}}\,dx$.

a) 2 b) 3 c) 4 d) 5 e) $2\sqrt{5}-1$
f) 9/2 g) 16/3 h) 17/4 i) 21/4 j) $3\sqrt{5}-1$

Solution: c

```
> int(1/sqrt(5-x),x = 1 .. 5);
```

$$4$$

16. Calculate $\int_0^{\frac{\pi}{2}} \frac{\cos(x)}{\sqrt{\sin(x)}}\,dx$.

a) 2 b) 3 c) 4 d) $\sqrt{2}$ e) $\frac{\sqrt{3}-1}{2}$

f) $\frac{\sqrt{3}}{2}$ g) $\frac{\sqrt{2}}{2}$ h) $\sqrt{2}-\frac{1}{2}$ i) $1-\frac{\sqrt{2}}{2}$ j) $\frac{\sqrt{2}-1}{2}$

Solution: a

```
> int(cos(x)/sqrt(sin(x)), x = 0 .. Pi/2);
```

$$2$$

17. Calculate $\int_0^{\infty} \frac{x}{(1+x^2)^{\left(\frac{3}{2}\right)}}\,dx$.

a) $\frac{1}{2}$ b) $\frac{2}{3}$ c) $\frac{3}{2}$ d) $\frac{3}{4}$ e) 1

f) $\frac{4}{3}$ g) $2\sqrt{2}$ h) $\frac{\sqrt{2}}{2}$ i) $\frac{\sqrt{2}-1}{2}$ j) 2

Solution: e

```
> int(x/((1+x^2)^(3/2)),x = 0 .. infinity);
```

$$1$$

```
>
>
```

18. It may be verified that $\frac{d}{dx}\left(\frac{\ln(x)}{x}\right) = \frac{1}{x^2} - \frac{\ln(x)}{x^2}$. **Hence or otherwise, evaluate**

$$\int_1^{\infty} \frac{1-\ln(x)}{x^2}\,dx.$$

a) 0 b) −1 c) 1 d) e e) $\frac{1}{e}$

f) e − 1 g) $\frac{e-1}{2}$ h) 2 e i) $\frac{2}{e}$ j) $\frac{e}{2}$

Solution: a

```
> eqn := Int((1-ln(x))/(x^2),x = 1 .. infinity) = Limit(subs(x=N, ln(x)/x ) - subs(x=1, ln(x)/x ), N = infinity);
```

$$eqn := \int_1^\infty \frac{1-\ln(x)}{x^2}\,dx = \lim_{N\to\infty} \frac{\ln(N)}{N} - \ln(1)$$

```
> eqn2 := lhs(eqn) = simplify( rhs(eqn) );
```

$$eqn2 := \int_1^\infty \frac{1-\ln(x)}{x^2}\,dx = \lim_{N\to\infty} \frac{\ln(N)}{N}$$

```
> eqn3 := lhs(eqn) = Limit(Diff(ln(N),N)/Diff(N,N), N = infinity); #L'Hopital
```

$$eqn3 := \int_1^\infty \frac{1-\ln(x)}{x^2}\,dx = \lim_{N\to\infty} \frac{\frac{d}{dN}\ln(N)}{\frac{d}{dN}N}$$

```
> eqn4 := lhs(eqn) = value( rhs( eqn3) );
```

$$eqn4 := \int_1^\infty \frac{1-\ln(x)}{x^2}\,dx = 0$$

19. Calculate the fifth partial sum of $\sum_{n=1}^{\infty} \ln\left(\frac{n+1}{n}\right)$.

a) $\ln(5)$ **b)** $\ln(6)$ **c)** $\ln\left(\frac{6}{5}\right)$ **d)** $\ln\left(\frac{5}{6}\right)$ **e)** $2\ln(5)$

f) $2\ln(6)$ **g)** $5\ln\left(\frac{6}{5}\right)$ **h)** $6\ln\left(\frac{6}{5}\right)$ **i)** $5\ln\left(\frac{5}{6}\right)$ **j)** $6\ln\left(\frac{5}{6}\right)$

Solution: b

```
> answer := sum( ln((n+1)/n), n = 1..5);
```

$$answer := \ln(2) + \ln\left(\frac{3}{2}\right) + \ln\left(\frac{4}{3}\right) + \ln\left(\frac{5}{4}\right) + \ln\left(\frac{6}{5}\right)$$

```
> answer := simplify( answer );
```

$$answer := \ln(2) + \ln(3)$$

```
> combine( answer , ln);
```

$$\ln(6)$$

20. Consider

I) $\sum_{n=1}^{\infty} \frac{1}{n}$ **II)** $\sum_{n=1}^{\infty} \sin\left(\frac{1}{n}\right)$ **III)** $\sum_{n=1}^{\infty} \cos\left(\frac{1}{n}\right)$ **IV)** $\sum_{n=1}^{\infty} \frac{n-1}{n}$

List all given series for which the Divergence Test yields a conclusion.

a) I b) II c) III d) IV e) I, II

f) I, III g) I, IV h) II, III i) II, IV j) III, IV

Solution: j

```
> a := n -> 1/n;
  b := n -> sin(1/n);
  c := n -> cos(1/n);
  d := n -> (n-1)/n;
```

$$a := n \rightarrow \frac{1}{n}$$

$$b := n \rightarrow \sin\left(\frac{1}{n}\right)$$

$$c := n \rightarrow \cos\left(\frac{1}{n}\right)$$

$$d := n \rightarrow \frac{n-1}{n}$$

```
> testeq( limit(a(n), n = infinity) = 0 );
```

true

```
> testeq( limit(b(n), n = infinity) = 0 );
```

true

```
> testeq( limit(c(n), n = infinity) = 0 ); #Divergence Test
  applies
```

false

```
> testeq( limit(d(n), n = infinity) = 0 ); #Divergence Test
```

`applies`

false